# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>

pkgbase=libdbusmenu-qt
pkgname=(libdbusmenu-qt5)
pkgver=0.9.3+16.04.20160218
pkgrel=5
pkgdesc="A library that provides a Qt implementation of the DBusMenu spec"
arch=(x86_64)
url="https://launchpad.net/libdbusmenu-qt"
license=(GPL)
makedepends=(cmake doxygen qt5-base)
source=("http://archive.ubuntu.com/ubuntu/pool/main/libd/$pkgbase/${pkgbase}_$pkgver.orig.tar.gz")
sha256sums=('a8e6358a31c44ccdf1bfc46c95a77a6bfc7fc1f536aadb913ed4f4405c570cf6')

prepare() {
  mkdir -p build
}

build() {
  cd build
  cmake ../$pkgbase-$pkgver \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DCMAKE_INSTALL_LIBDIR=lib
  make
}

package_libdbusmenu-qt5() {
  depends=(qt5-base)

  cd build
  make DESTDIR="$pkgdir" install
}
